



DATO OG KLOKKESLÆT

Funktioner i Excel

Indholdsfortegnelse

Sådan håndtere Excel datoer og klokkeslæt	3
Forstå datoserienumre	3
Vælg dit datasystem: 1900 eller 1904.....	3
Indtastning af datoer	4
Søg efter datoer.....	6
Forstå tidsserienumre	6
Indtast klokkeslæt	7
Formater datoer og klokkeslæt	8
Problemer med datoer	10
Excel skudårs fejlen	10
Datoer før 1900	10
Inkonsekvente dato indtastninger.....	11
Dato relaterede funktioner	11
Vis dagsdato	12
Vis en dato med en funktion	13
Generer en datoserie.....	13
Konverter en ikke datostreng til en dato	14
Beregn antallet af dage mellem to datoer	15
Beregn antallet af arbejdsdage mellem to datoer	16
Modregn en dato ved kun at bruge arbejdsdage.....	16
Beregn antallet af år mellem to datoer	17
Beregn en persons alder.....	17
Bestem hvilken dag det er i året.....	18
Hvor er funktionen DATO.FORSKEL?	18
Bestem dagen i ugen	19
Bestem ugen i året.....	19
Bestem datoen for den seneste søndag.....	19
Bestem den første dag i ugen efter en dato.....	19
Bestem den n'te forekomst af en ugedag i en måned	20
Tæl antallet af forekomster af en ugedag	20
Bestem den sidste dag i en måned.....	21
Bestem om år er et skudår	21
Bestem en datos kvartal.....	22

Konverter et år til romerske tal	22
Tidsrelaterede funktioner.....	22
Få vist en nuværende tid	23
Får vist klokkeslæt ved hjælp af en funktion.....	23
Beregn forskellen mellem to klokkesæt	24
Sammentælling af klokkeslæt der overstiger 24 timer	25
Konverter fra militær klokkeslæt.....	27
Konverter decimal timer, minutter eller sekunder til et klokkeslæt.....	27
Tilføj timer, minutter eller sekunder til et klokkeslæt.....	27
Konverter mellem tidszoner	28
Afrunding af klokkeslæt.....	29
Arbejdet med tidsværdier	29

Datoer og klokkeslæt

I dette notesæt

- En oversigt over arbejdet med datoer og klokkeslæt i Excel
- Excels dato relaterede funktioner
- Excels klokkeslæts relaterede funktioner

Nybegyndere synes ofte at arbejdet med datoer og klokkeslæt i Excel kan være frustrerende. For at undgå disse frustrationer, skal du have en god forståelse for hvordan Excel håndtere tidsbaseret informationer. Dette notesæt giver dig de informationer du har brug for, for at oprette kraftfulde formler, der manipulerer datoer og klokkeslæt.

Datoerne i dette notesæt korresponderer til det danske datoformat: dag-måned-år. F.eks. 03-12-2017 refererer til den 3. december 2017 og ikke den 12. marts 2017. Jeg ved at nogle vil finde det ulogisk, men der er mange forskellige opfattelser af hvordan datoer skrives.

Sådan håndtere Excel datoer og klokkeslæt

I dette afsnit præsenteres en hurtig oversigt over hvordan Excel håndtere datoer og klokkeslæt, det inkluderer en gennemgang af Excels dato og klokkeslæts serienummersystem, og giver også tips til indtastning og formatering af datoer og klokkeslæt.

Forstå datoserienumre

For Excel er en dato ganske enkelt et tal. Mere præcist et serienummer, der repræsenterer antallet af dage siden 0. januar 1900. Serienummeret 1 korresponderer til den 1. januar 1900, serienummeret 2 korresponderer til den 2. januar 1900 osv. Dette system gør det muligt at oprette formler, der udfører beregninger med datoer. F.eks. kan du oprette en formel, der beregner antallet af dage mellem to datoer, eller bestemme datoen på den 3. fredag i januar 2045.

Du undere dig måske over den 0. januar 1900. Denne ikke eksisterende dato (som korresponderer med serienummeret 0) bruges faktisk til at repræsentere tider, der ikke associeres med en bestemt dag. Det bliver klart senere i dette notesæt.


Excel 2000 og senere versioner understøtter datoer fra den 1. januar 1900 til den 31. december 9999 (serienummer 2.958.465). Versioner før Excel 2000 understøtter et meget mindre interval af datoer: fra den 1. januar 1900 til den 31. december 2078 (serienummer 65.380).

Vælg dit datosystem: 1900 eller 1904

Excel understøtter faktisk to datosystemer: 1900 datosystemet og 1904 datosystemet. Hvilket system du bruger i en projektmappe afgør hvilken dato, der fungerer som en basis for datoerne. Datosystemet 1900 benytter den 1. januar 1900, som den dato der har datoserienummeret 1. Datosystemet 1904 benytter den 1. januar 1904 som det basisdato. Som standard benytter Excel til Windows datosystemet 1900, og Excel til Mac brugere benytter datosystemet 1904 – eller det har det i hvert fald gjort. Microsoft har lavet en

ændring, og nu benytter Excel 2011 til Mac (og sandsynligvis efterfølgende version) datosystemet 1900 som standard.

Excel til Windows understøtter datosystemet 1904 for at være kompatibelt med Mac filer. Du vælger datosystemet 1904 i Excels dialogboks Indstillinger. (vælg Filer → Indstillinger og find frem til Brug 1904-datosystem under fanen Avanceret.)



Normalt bør du bruge standard datosystemet 1900. Og du bør udvise forsigtighed, hvis du bruger to forskellige datosystemer i projektmapper, der er forbundne. Antag f.eks. at Regneark 1 benytter datosystemet 1904 og indeholder datoen 15. januar 1999 i celle A1. Antag endvidere at Regneark 2 benytter datosystemet 1900 og indeholder et link til celle A1 i Regneark 1. Regneark 2 vil vise datoen som 14. januar 1995. Begge regneark benytter det samme datoserienummer (34713), men de fortolkes forskelligt.

En fordel ved at benytte datosystemet 1904 er at det giver dig mulighed for at vise negative klokkeslæt. Med datosystemet 1904, vil en beregning, der resultere i en negativ tid vises som -1:30: det er en forskel på 1 time og 30 minutter.

Indtastning af datoer

Du kan indtaste en dato direkte som et serienummer (hvis du kender det), men ofte vil du vælge at indtaste en dato ved hjælp af et af flere genkendelige datoformater. Excel vil automatisk konvertere din indtastning til det korresponderende datoserienummer (som benyttes til beregninger), og samtidig tilføje et datoformat til cellen, så det vises som en letlæselig dato i stedet for et kryptisk serienummer.

Hvis du f.eks. har brug for at indtaste 20 august 2017, kan du indtaste datoen ved at skrive **20 august 2017** (eller bruge et af de mange forskellige datoformater). Excel fortolker din indtastning og gemmer værdien

42967, som er datoserienummeret for denne dato. Excel tilføjer også et af de mange datoformater afhængig af hvordan datoen oprindelig er indtastet, så celleindholdet ser måske ikke helt ud på samme måde som du indtastede det.

Afhængig af dine regionale indstillinger, vil indtastningen af en dato i et format som 20 august 2017 måske blive opfattet som en tekststreng. I dette tilfælde, skal du indtaste datoen i et format, der korresponderer med din regionale indstilling, som f.eks. august 20, 2017.

Når du aktiverer en celle der indeholder en dato, vil Formellinjen vise celleindholdet formateret med standard datoformatet – som korresponderer med dit systems korte datoformat. Formellinjen viser *ikke* datoens serienummer – som er inkonsistent med andre datatyper af talformateringer. Hvis du har brug for at undersøge serienummeret for en bestemt dato, formater da cellen med formatet Standard.

Tabel 1 viser en oversigt over forskellige datoformater Excel genkender (ved at bruge den Danske indstilling). Resultatet kan variere, hvis du bruger en anden regionalindstilling. De viste datoer antager at cellen ikke indeholder en numerisk formatering.

Tabel 1: Data indtastnings formater genkendt af Excel

Indtastning	Excels fortolkning (Dansk indstilling)	Hvad Excel viser
20-8-2017	20 august 2017	Windows korte datoformat
20-08-2017	20 august 2017	20-08-2017
20/8/17	20 august 2017	Windows korte datoformat
20/8/2017	20 august 2017	Windows korte datoformat
20/8-17	20 august 2017	Windows korte datoformat
20 august 2017	20 august 2017	20-aug-17
Aug 20	20 august i det nuværende år	aug-20
August 20	21 august i det nuværende år	aug-20
August20	22 august i det nuværende år	aug-20
20-aug-2017	20 august 2017	20-aug-17
2017-8-20	20 august 2017	Windows korte datoformat

Som du kan se i tabel 1 er Excel ganske god til at genkende data indtastet i en celle. Den er dog ikke perfekt. F.eks. kan Excel ikke genkende de følgende indtastning som datoer:

- August 20 2017
- August-20 2017
- August-20/2017

I stedet bliver disse indtastninger opfattet som tekst. Hvis du har planer om at bruge datoer i formler, skal du sikre dig at Excel kan genkende datoen du indtaster som en dato. Ellers vil formlerne, der refererer til disse datoer, generere et fejlsresultat.

Hvis du forsøger indtaster en dato, der ligger uden for det datoområde der understøttes, vil Excel opfatte det som en tekst. Hvis du forsøger at formatere et serienummer, der ligger uden for det datoområde, der understøttes, vil værdien der vises som en række havelåger (#####).

Hvis du vil være helt sikker på at en gyldig dato er indtastet i en bestemt celle, kan du bruge Excel dato validerings værktøj. Du skal bare angive et validering kriterie der kun tillader datoer. Du kan også specificere et bestemt datointerval.

Søg efter datoer

Hvis dit regneark benytter mange datoer, kan du have brug for at søge efter en bestemt dato ved at benytte dialogboksen Søg og vælg (Hjem → Søg og vælg → Søg eller tast Ctrl + B). Excel er ret kræsen når det drejer sig om at finde datoer. Du skal indtaste datoen som den bliver vist i Formellinjen. F.eks. hvis en celle indeholder en dato formateret så den bliver vist som 20 august 2017, vil datoen i Formellinjen blive vist med dit systems korte datoformat (f.eks. 20-08-2017). Hvis du derfor søger efter datoen som den vises i cellen vil Excel ikke finde den. Den vil derimod finde cellen, hvis du søger efter datoen i formatet, der vises i Formellinjen.

Forstå tidsserienumre

Når du har brug for at arbejde med klokkeslæt, udvider du ganske enkelt Excels datoserienummer system til at inkludere decimaler. Med andre ord, vil Excel arbejde med klokkeslæt ved at bruge brøkdage. F.eks. er datoserienummeret for datoen 20-08-2017 42967. Middag (halvvejs gennem dagen) repræsenteres internt som 42967,5.

Serienummeret der svare til et minut er ca. 0,0006944. Formlen der følger beregner dette tal ved at gange 24 timer med 60 minutter og så divideres resultat op i 1. Nævneren består af antallet af minutter i en dag (1.440).

```
=1 / (24 * 60)
```

På samme måde er serienummeret ækvivalent til 1 sekund ca. 0,00001157. Det kan fås ved følgende formel (1 delt med 24 timer ganget med 60 minutter ganget med 60 sekunder). I dette tilfælde, repræsenterer nævneren antallet af sekunder på en dag (86.400).

```
=1 / (24 * 60 * 60)
```

I Excel er den mindste tidsenhed en tusinddel af et sekund. Tidsserienummeret vist her repræsenterer 23:59:59,999 eller en tusinddel af et sekund før midnat:

```
0,9999999884259
```

Tabel 2 viser forskellige tider på dagen, sammen med deres tilhørende tidsserienummer.

Tabel 2: Tider på dagen og deres tilhørende serietal

Tid på dagen	Tidsserietal
00:00:00 (midnat)	0,0000
01:30:00	0,0625
03:00:00	0,1250
04:30:00	0,1875
06:00:00	0,2500
07:30:00	0,3125
09:00:00	0,3750
10:30:00	0,4375
12:00:00 (middag)	0,5000
12:30:00	0,5208
15:00:00	0,6250
16:30:00	0,6875
18:00:00	0,7500
19:30:00	0,8125
21:00:00	0,8750
22:30:00	0,9375

Indtast klokkeslæt

Som med indtastning af datoer, behøver du normalt ikke bekymre dig om det faktiske tidsserietal. Indtast bare klokkeslættet i en celle ved at bruge et genkendeligt format. Tabel 3 viser eksempler på klokkeslæt formateret som Excel genkender.

Tabel 3: Klokkeslæt indtastningsformater genkendt af Excel

Indtastning	Excels fortolkning	Hvad Excel viser
11:30:00	11:30:00	11:30:00
11:30:00 AM	11:30:00	11:30:00 AM
11:30:00 PM	23:30:00	11:30:00 PM
11:30	11:30:00	11:30
13:30	13:30:00	13:30
11:00 AM	11:00:00	11:00 AM

Da de forgående eksempler ikke har en specifik dato der associerer dem, vil Excel bruge en dato med serietallet 0, som korresponderer til den ikke eksisterende dato 0. januar 1900.

Hvis du benytter datosystemet 1904, vil klokkeslætsværdier uden en bestemt dato benytte 1. januar 1904, som dets dato. Diskussionen der følger vil antage at du bruger standard datosystemet 1900.

Ofte vil du kombinere en dato og et klokkeslæt. Gør det ved at bruge et genkendeligt datoindtastningsformat, efterfulgt af et mellemrum, og så et genkendeligt klokkeslætsindtastningsformat. F.eks. hvis du indtaster følgende tekst i en celle, vil Excel opfatte det som 11:30 den 20. august 2017. Dets serietal er 42967,47917.

20-08-2017 11:30

Når du indtaster et klokkeslæt, der overskrider 24 timer, vil den associerede dato for klokkeslættet forøges tilsvarende. Hvis du f.eks. indtaster følgende klokkeslæt i en celle vil det blive opfattet som 1:00 den 1. januar 1900. Dagen skille ved forøgelsen af indtastningen, da tiden overskrider 24 timer. (Husk på at en tidsværdi indtastet uden en dato benytter den 0. januar 1900, som dets dato.)

25:00:00

Hvis du kun indtaster et klokkeslæt (uden en tilhørende dato) vil du opdage at det højeste klokkeslæt du kan indtaste i en celle er 9999:59:59 (lige under 10.000 timer). Excel tilføjer det nøjagtige antal dage. I dette tilfælde opfattes 9999:59:59 som 3:59:59 den 19. februar 1901. Hvis du indtaster et klokkeslæt, der overskrider 10.000 timer, vil klokkeslættet vises som en tekststreng.

Formater datoer og klokkeslæt

Du har en stor fleksibilitet i formateringen af celler, der indeholder datoer og klokkeslæt. F.eks. kan du formatere cellen til kun at vise en del af datoen, kun klokkeslættet, eller både dato og klokkeslæt.


En måde at formatere dato og klokkeslæt er ved at vælge cellerne og så bruge Talformater fra Hjem → Talformater (se figur 1). Talformater tilbyder dig to datoformater og et klokkeslæts format. Datoformaterne er dem der er specificeret som dit Windows Kort datoformat og Lang datoformat.

Når du opretter en formel der referere til en celle, der indeholder en dato eller klokkeslæt, vil Excel måske automatisk formatere formelcellen som en dato eller klokkeslæt. Nogle gange er dette meget hjælpsomt; andre gange er det fuldstændigt upassende og decideret irriterende. Desværre kan du ikke slå denne automatiske datoformatering fra. Du kan dog benytte en smart genvejstast til at fjerne al talformatering fra cellen og vende tilbage til standard formateringen Generel. Du vælger bare cellen og taster Ctrl+Shift+~.


For at få mere kontrol over dato og klokkeslæt formateringen, markeres cellerne, og gør da brug af fanen Tal i dialogboksen *Formatér celler*, som vist i figur 2. Her er forskellige måder at få vist denne dialogboks:

- Klik på ikonet start dialogboks i gruppen Tal på fanen Hjem.
- Klik i listefeltet Talformater og vælg Flere talformater i listen der dukker op.
- Tast Ctrl + 1.

Dato og klokkeslæt i Excel



Figur 1 Benyt listefeltet Talformater for at ændre udseendet på datoer og klokkeslæt.



Figur 2 Benyt fanen Tal i dialogboksen Formatér celler for at ændre udseendet på datoer og klokkeslæt.

Kategorien Dato viser alle de indbyggede formater, og kategorien Klokkeslæt viser alle de indbyggede klokkeslæt formater. Yderligere dato og klokkeslæt formater er tilgængelige i kategorien Brugerdefineret.

Nogle formater inkludere både visning af dato og klokkeslæt. Vælg det ønskede format i listen Type og klik OK.

Bemærk at de to første datoformater i kategorien Dato og de to første formater i kategorien Klokkeslæt korresponderer med din Windows dato og klokkeslæt indstillinger.

Hvis ingen af de indbyggede formater lever op til dine ønsker, kan du oprette et brugerdefineret talformat. Vælg kategorien Brugerdefineret og skriv dit brugerdefinerede format i feltet Type.

Problemer med datoer

Excel har nogle problemer når det drejer sig om datoer. Mange af disse problemer stammer fra det faktum at Excel er designet for mange år side, før forkortelsen Y2K blev et velkendt udtryk. Excel designerne efterlignede ganske enkelt Lotus 1-2-3s begrænsede dato og klokkeslæts funktioner, som indeholder en grim fejl der blev kopieret over i Excel med vilje. Udeover dette, har versioner af Excel udvist inkonsistens i hvordan de fortolker en celleindtastning, der har et to cifret årstal. Og endelig hvordan Excel opfatter en datoindtastning afhænger af de regionale datoindstillinger.

Hvis Excel skulle designes fra bunden i dag, er jeg sikker på at det ville blive meget mere alsidigt i håndteringen af datoer. Desværre er vi for tiden bundet af et produkt, der lader meget at ønske i området datoer.

Excel skudårs fejlen

Et skudår, som kommer hver fjerde år, indeholder en ekstra dag (den 29. februar). Specielt år der kan deles med 100 er ikke skudår, med mindre de også kan deles med 400. Selvom år 1900 ikke er et skudår, opfatter Excel det som værende et skudår. Med andre ord når du taster følgende i en celle, klager Excel ikke. Det opfatter det som en gyldig dato og tildeler det et serietallet 60:

```
29-02-1900
```

Hvis du indtaster den følgende ugyldige dato, vil Excel helt korrekt opfatte det som en fejl og ikke konvertere det til en dato. I stedet laver den celle indholdet om til en tekststreng:

```
29-02-1901
```

Hvordan kan et produkt der benyttes af millioner af mennesker indeholder en fejl der er så åbenlys? Svaret er historisk. Den originale version af Lotus 1-2-3 indeholdte en fejl, der bevirkede at det opfattede 1900 som et skudår. Da Excel blev udsendt noget tid senere, kendte designerene til denne fejl, og valgte at føre den videre til Excel for at vedligeholde kompatibiliteten med Lotus regneark filerne.

Hvorfor eksistere denne fejl stadig i senere versioner af Excel? Microsoft vurderede at ulemperne ved at rette denne fejl overskyggede fordelene. Hvis fejlen blev fjernet ville det skabe problemer i tusindvis af eksisterende regneark. Ud over dette ville rettelsen af dette problem påvirke kompatibiliteten mellem Excel og andre programmer, der benytter datoer. Som det er nu skaber denne fejl meget få problem, da de færreste brugere benytter datoer før den 1. marts 1900.

Datoer før 1900

Verden begyndt selvfølgelig ikke den 1. januar 1900. Folk som benytter Excel i deres arbejde med historiske informationer har ofte brug for at arbejde med datoer, der ligger før den 1. januar 1900. Desværre er den eneste måde at arbejde med datoer før 1900, at indtaste dem i en celle som tekst. F.eks. kan du indtaste følgende i en celle, og Excel vil ikke gøre vrøvl:

18 april 1885

Hvis du har plane om at sortere informationer med ældre datoer indtastet som tekst, bør du indtaste din tekstdatoer med fire cifre til årstallet, efterfulgt af to cifre til måneden og så to cifre til dagen. Det kan gøre sådan: 1885-04-18. Dette format vil gøre det muligt at udføre en præcis sortering.

Du kan dog ikke udføre manipulationer med en dato, der genkendes som en tekst. F.eks. kan du ikke ændre dets numeriske formatering, du kan ikke afgøre hvilken dag i ugen datoen optræder, og du kan ikke beregne datoen, der kommer syv dage senere.

Inkonsekvente dato indtastninger

Du skal være forsigtig når du indtaster datoer, der bruger to cifre til årstallet. Når du gør dette har Excel nogle regler der går i gang, for at afgøre hvilket århundrede der skal bruges. Disse regler varierer, afhængig af hvilken version af Excel du bruger.

Tocifret årstal mellem 00 og 29 fortolkes som datoer i det 21. århundrede, og tocifret årstal mellem 30 og 99 fortolkes som datoer i det 20. århundrede. F.eks. hvis du indtaster 15-12-28, vil Excel fortolke din indtastning som den 15. december 2028. Men hvis du indtaster 15-12-30, vil Excel se det som den 15. december 1930, da Windows som standard har en årsgrænse på 2029. Du kan beholde denne standard eller du kan ændre den i Windows kontrolpanelet.

Den bedste måde at undgå overraskelser på er ganske enkelt at indtaste alle år med fire cifre.

Dato relaterede funktioner

Excel har en hel del funktioner, der arbejder med datoer. De er alle at finde i listefeltet under Dato og klokkeslæt i gruppen Formler → Funktionsbiblioteket.

Tabel 4 opsummerer de datorelaterede funktioner, der er tilgængelige i Excel.

Tabel 4: Datorelaterede funktioner

Funktion	Beskrivelse
DATE	Returnerer serienummeret for en bestemt dato
DATO.FORSKEL	Beregner antallet af dage, måneder eller år mellem to datoer. Denne funktion er nyttig i formler, hvor du skal beregne en alder.
DATOVÆRDI	Konverterer en dato i form af tekst til et serienummer
DAY	Konverterer et serienummer til en dag i måneden
DAGE (2013)	Returnerer antallet af dage mellem to datoer
DAGE360	Beregner antallet af dage mellem to datoer på grundlag af et år med 360 dage
EDATO *	Returnerer serienummeret for den dato, der ligger det angivne antal måneder før eller efter startdatoen
SLUT.PÅ.MÅNED *	Returnerer serienummeret på den sidste dag i måneden før eller efter et angivet antal måneder
TID	Konverterer et serienummer til en time

Funktion	Beskrivelse
ISOUGE.NR ***	Returnerer tallet for ISO-ugenummeret for årstallet for en given dato
MINUT	Konverterer et serienummer til et minut
MÅNED	Konverterer et serienummer til en måned
ANTAL.ARBEJDSDAGE *	Returnerer antallet af hele arbejdsdage mellem to datoer
ANTAL.ARBEJDSDAGE.INTL **	Returnerer antallet af hele arbejdsdage mellem to datoer ved hjælp af parametre for at angive, hvilke og hvor mange dage der er weekenddage
NU	Returnerer serienummeret for den aktuelle dato eller det aktuelle klokkeslæt
SEKUND	Konverterer et serienummer til et sekund
TID	Returnerer serienummeret for et bestemt klokkeslæt
TIDSVÆRDI	Konverterer et klokkeslæt i form af tekst til et serienummer
I DAG	Returnerer serienummeret for dags dato
UGEDAG	Konverterer et serienummer til en ugedag
UGE.NR *	Konverterer et serienummer til et tal, der angiver ugenummeret i året
ARBEJDSDAG *	Returnerer serienummeret for dagen før eller efter det angivne antal arbejdsdage
ARBEJDSDAG.INTL **	Returnerer serienummeret for datoen før eller efter et angivet antal arbejdsdage ved hjælp af parametre for at angive, hvilke og hvor mange dage der er weekenddage
ÅR	Konverterer et serienummer til et år
ÅR.BRØK *	Returnerer årsbrøken, der repræsenterer antallet af hele dage mellem startdato og slutdato

* I versioner før Excel 2007, var denne kun tilgængelig med Analysis ToolPak installeret

** En funktion der blev introduceret i Excel 2010

*** En funktion der blev introduceret i Excel 2013

Vis dagsdato

Den følgende funktioner viser dagsdato i en celle:

```
=IDAG ()
```

Du kan vise datoen kombineret med tekst. Den følgende formel er et eksempel, der viser teksten *I dag er det søndag 30. juli 2017:*

```
= "I dag er det "&TEKST (IDAG () ; "dddd d. mmmm åååå")
```

Det er vigtigt at huske på at funktionen IDAG opdateres hver gang at regnearket genberegnes. Hvis du f.eks. indtaster den forgående formel i et regneark, viser formelen dagsdato. Når du åbner regnearket i morgen, vil den vise dagsdato for den dag (ikke datoen da du indtastede formelen).

For at indtaste et datostempel i en celle taster du Ctrl+; (semiklon). Dette indsætter datoen direkte i en celle og gør ikke brug af en formel. Datoen vil derfor ikke skifte.

Vis en dato med en funktion

Som forklaret tidligere i notesættet, kan du nemt indtaste en dato i en celle ved at skrive den, ved hjælp af et af datoformaterne som Excel genkender. Du kan også oprette en dato ved hjælp af funktionen DATO, som har tre argumenter: år, måned og dag. F.eks. vil den følgende formel returnere en dato, der omfatter året i celle A1, måneden i celle B1 og dagen i celle C1:

```
=DATO (A1 ; B1 ; C1)
```

Funktionen DATO acceptere ugyldige argumenter og justere resultatet derefter. F.eks. benytter den næste formel 13 som argument for måned, og returnere 1. januar 2018. Argumentet for måned overføres automatisk til den første måned i det efterfølgende år.

```
=DATO (2017 ; 13 ; 1)
```

Ofte vil du bruge DATO funktionen med andre funktioner som argumenter. F.eks. formelen der følger bruger funktionerne ÅR og IDAG til at returnere datoen for juleaften (24. december) i det pågældende år:

```
=DATO (ÅR (IDAG ()) ; 12 ; 24)
```

Funktionen DATOVÆRDI konvertere en tekststreng, der ligner en dato til et datoserietal. Den følgende formel returnere 43093, datoserietallet for 24. december 2017:

```
=DATOVÆRDI ("24-12-2017")
```

For at se resultatet som en dato, er du nødt til at tilføje en dato formatering til cellen.

Vær forsigtig når du bruge funktionen DATOVÆRDI. En tekststreng der ligner en dato i dit land er måske ikke den samme i et andet land. De forgående eksempler virker hvis dit system er sat op til Dansk datoformatering, men den vil returnere en fejl i andre regionale datoformateringer, da Excel måske ser efter den 12. i den 24. måned!

Generer en datoserie

Ofte vil du indsætte en serie af datoer i et regneark. F.eks. hvis du vil spore et ugentligt salg, kan du have interesse i at indsætte en serie af datoer, hver adskilt med syv dage. Disse datoer skal identificere salgstallene.

I nogle tilfælde kan du bruge Excels autoudfyldning til at indsætte en serie af datoer. Indtast den første dato og træk cellens kopieringshåndtag, men du holder højre museknappen nede. Slip museknappen og vælg en indstilling i genvejsmenuen (se figur 3) – Fyld dage, Fyld ugedage, Fyld måneder eller Fyld år. Bemærk at Excel ikke tilbyder indstillingen Fyld uge.

For at få mere fleksibilitet kan du indtaste de to første dage i serien – f.eks. startdatoen for uge 1 og start datoen for uge 2. Marker begge celler og trække kopieringshåndtaget ned over kolonnen. Excel vil udfylde datoserien, hvor hver dato er adskilt af et interval repræsenteret af de to første datoer.

Fordelen ved at bruge formler (i stedet for funktionen Autoudfyldning) når en serie skal oprettes er at du kan ændre den første dato, og de andre vil da automatisk opdateres. Du skal indtaste startdatoen i en celle og derefter bruge formler (kopieret ned over kolonne) for at generere yderligere datoer.


Det følgende eksempel antager at du har indtastet den første dato i serien i celle A1, og formelen i celle A2. du kan derefter kopiere formelen ned over kolonnen lige så mange gange du har brug for.

For at generere en serie af datoer adskilt af syv kan du bruge denne formel:

```
=A1+7
```

For at generere en serie af datoer der er adskilt med en måned, har du brug for en mere kompliceret formel, da måneder ikke altid har det samme antal dage. Denne formel opretter en serie af datoer, adskilt med en måned:

```
=DATO (ÅR (A1) ; MÅNED (A1) +1 ; DAG (A1) )
```



Figur 3 Benyt Excels Autoudfyldnings funktion til at oprette en serie af datoer.

For at generere en serie af datoer der er adskilt med et år kan du benytte denne formel:

```
=DATO (ÅR (A1) +1 ; MÅNED (A1) ; DAG (A1) )
```

For at generere en serie af ugedage (ingen lørdage og søndage) kan du benytte følgende formel der følger. Denne formel antager at celle A1 ikke er en lørdag eller søndag:

```
=HVIS (UGEDAG (A1) =6 ; A1+3 ; A1+1)
```

Konverter en ikke datostreng til en dato

Du skal måske importere data, der indeholder datoer kodet som tekststreng. F.eks. følgende tekst, der repræsenterer 19 august 2017 (firecifret årstal efterfulgt af en tocifret måned, fulgt af en tocifret dag):

```
20170819
```

For at konvertere denne streng til en egentlig dato, kan du bruge en formel som denne, der antager at den kodet dato er i celle A1:

```
=DATO (VENSTRE (A1; 4) ; MIDT (A1; 5; 2) ; HØJRE (A1; 2) )
```

Denne formel benytter tekstfunktioner (VENSTRE, MIDT og HØJRE) til at udtrække cifrene, og derefter bruger de udtrukne cifre som argumenter i DATO funktionen.

Flere informationer kan fås i notesættet om hvordan du bruger formler til at manipulere med tekster.

Beregn antallet af dage mellem to datoer

En almindelig type af beregning bestemmer antallet af dage mellem to datoer. F.eks. har du et økonomiregnskab der beregner renter, der er optjent på en opsparingskonto. Renterne afhænger af hvor mange dage kontoen er åben. Hvis dit regneark indeholder startdatoen og slutdatoen for din konto, kan du beregne antallet af dage kontoen var åben.

Da datoer er gemt som fortløbende serietal kan du benytte en simpel subtraktion til at beregne antallet af dage mellem to datoer. F.eks. hvis cellerne A1 og B1 begge indeholder en dato, vil følgende formel returnere antallet af dage mellem disse datoer:

```
=A1-B1
```

Hvis celle B1 indeholder en senere dato end datoen i celle A1, vil resultatet blive negativ. Hvis du ikke bekymrer dig om hvilken dato der er den første og vil undgå, at der bliver vist en negativ værdi, kan du bruge denne formel:

```
=ABS (A1-B1)
```

Du kan også bruge regnearksfunktionen DAGE, introduceret i Excel 2013. Den tilbyder ikke som jeg lige umiddelbart kan se det nogen fordele, men her er et eksempel på hvordan man kan bruge den til at beregne antallet af dage mellem to datoer:

```
=DAGE (A1; B1)
```

Det kan undertiden være mere besværligt at beregne forskellen mellem to datoer. For at demonstrere dette kan du betragte analogien om "Hegnspæle". Hvis du bliver spurgt om hvor mange enheder der skal til for at lave et hegn, kan du komme med to svar: antallet af pæle eller antallet af mellemrum mellem pælene. Antallet af hegnspæle er altid en mere end antallet af mellemrum mellem hegnspælene.

For at overføre denne analogi til datoer, forestil dig at starten på en udsalgs periode er den 1. februar og udsalget slutter den 9. februar. Hvor mange dage varer udsalget? Hvis du trækker 1. februar fra 9. februar er svaret otte. Men udsalget varer rent faktisk ni dage. I dette tilfælde er det rigtige svar antallet af hegnspæle, og ikke mellemrummene. Formlen der skal til for at beregne længden på udsalget (forudsat du har de dertil hørende navngivet celler) ser sådan ud:

```
=SlutDag-StartDag+1
```

Beregn antallet af arbejdsdage mellem to datoer

Når du skal beregne forskellen mellem to datoer, vil du måske udelade weekender og helligdage. F.eks. har du brug for at vide hvor mange arbejdsdage der er i marts måned. Denne beregning skal udelukke lørdag, søndag og helligdage. Det kan her være en hjælp at bruge funktionen ANTAL.ARBEJDSDAGE.

Funktionen ANTAL.ARBEJDSDAGE beregner forskellen mellem to datoer og udelukker weekender (lørdag og søndag). Som en mulighed kan du specificere området af celler, der indeholder datoer på helligdage, som du yderligere ønsker at inkludere. Excel har ingen mulighed for at afgøre hvilke dage der er helligdage, så du er nødt til at angive denne information i et område.

Figur 4 viser et regneark der beregner antallet af arbejdsdage mellem to datoer. Området A2:A10 indeholder en liste med helligdag. Formlen i kolonne C beregner arbejdsdagene mellem datoer i kolonne A og kolonne B. Formlen C14 er:

```
=ANTAL.ARBEJDSDAGE (A14;B14;A2:A10)
```

	A	B	C
1	Dato	Helligdag	
2	01-01-2018	1. nytårsdag	
3	25-03-2018	Palmesøndag	
4	29-03-2018	Skærtorsdag	
5	30-03-2018	Langfredag	
6	01-04-2018	Påskedag	
7	02-04-2018	2. påskedag	
8	27-04-2018	Store bededag	
9	10-05-2018	Kristi himmelfartsdag	
10	20-05-2018	Pinsedag	
11	21-05-2018	2. pinsedag	
12			
13			
14	Første dag	Sidste dag	Arbejdsdag
15	mandag 01-01-2018	søndag 01-07-2018	123
16	mandag 01-01-2018	søndag 01-04-2018	62
17			

Figur 4 Benyt funktionen ANTAL.ARBEJDSDAGE til at beregne antallet af arbejdsdage mellem to datoer.

Denne formel returnerer 123, hvilket betyder at der i periode med start den 1. januar og slut den 1. juli er 123 arbejdsdage. Med andre ord vil beregningen udelukke helligdage og weekender. Formelen i celle C16 beregner antallet af arbejdsdag fra den 1. januar og frem til 1. april.

Excel 2010 introducerede en opdaterede version af funktionen ANTAL.ARBEJDSDAGE, der hed ANTAL.ARBEJDSDAGE.INTL. Denne nye version er brugbar, hvis du overvejer at gøre nogle andre dage end lørdag og søndag til weekend dage.

Modregn en dato ved kun at bruge arbejdsdage

Funktionen ARBEJDSDAGE er den modsatte af funktionen ANTAL.ARBEJDSDAGE. Hvis du f.eks. starter et projekt den 8. marts og projektet skal være færdigt 10 arbejdsdage senere, kan funktionen ARBEJDSDAGE beregne datoen for hvornår du skal have afsluttet projektet.

Den følgende formel benytter funktionen ARBEJDSDAGE til at bestemme datoen ti arbejdsdage efter den 8. marts 2017. En arbejdsdag er en ugedag (mandag til fredag).

```
=ARBEJDS DAG ("08-03-2018";10)
```

Formlen returnerer et datoserietal, som skal formateres til en dato. Resultatet er den 22. marts 2017 (der var fire weekend datoer mellem den 8. og 22 marts).

Den forgående formel kan returnere forskellige resultater afhængig af din regionale dato indstilling. En bedre formel er

```
=ARBEJDS DAG (DATO (2017;3;8);10)
```

Det andet argument i funktionen ARBEJDS DAG kan være negativt. Og som med funktionen ANTAL.ARBEJDS DAGE, accepterer funktionen ARBEJDS DAG et tredje argument (en reference til et område der indeholder en liste med helligdage).

Excel 2010 introducerede en opdaterede version af ARBEJDS DAG, kaldet ARBEJDS DAG.INTL. Denne nye version er brugbar, hvis du overvejer at gøre nogle andre dage end lørdag og søndag til weekend dage.

Beregn antallet af år mellem to datoer

Følgende formel beregner antallet af år mellem to datoer. Formlen antager at cellerne A1 og B1 begge indeholder datoer:

```
=ÅR (A1) -ÅR (B1)
```

Denne formel bruger funktionen ÅR til at udtrække årstallet fra hver dato og derefter trække det ene år fra det andet. Hvis celle B1 indeholder en senere dato end celle A1, vil resultatet være negativt.

Bemærk at denne funktion ikke beregner fulde år. Hvis f.eks. A1 indeholder 31-12-2017 og B1 indeholder 12-01-2018 vil formelen returnere en difference på et år selvom datoerne kun har en forskel på en dag.

Du kan også bruge funktionen ÅR.BRØK til at beregne antallet af år mellem to datoer. Denne funktion returnere antallet af år, inklusiv brøkdele af et år. F.eks.:

```
=ÅR . BRØK (A1;B1;1)
```

Da funktionen ÅR.BRØK ofte bliver brugt til finansielle programmer, benytter den et tredje valgfrit argument, der repræsenterer "basis" for året (f.eks. et år med 360 dage). Et tredje argument på 1 indikerer et faktisk år.

Beregn en persons alder

En persons alder indikerer antallet af hele år personen har levet. Formlen i den forgående sektion (til beregning af antallet af år mellem to datoer) vil ikke beregne denne værdi korrekt. Du kan dog benytte to andre formler til beregning af en persons alder.

Den følgende formel returnerer alderen på en person, hvor du indsætter fødselsdatoen i celle A1. Denne formel benytter funktionen ÅR.BRØK:

```
=HELTAL (ÅR . BRØK (IDAG ();A1;1))
```

Den følgende formel benytter funktionen DATO.FORSKEL til at beregne en alder. (Se afsnittet "Hvor er funktionen DATO.FORSKEL?")

```
=DATO.FORSKEL (A1; IDAG (); "y")
```

Bestem hvilken dag det er i året

Den 1. januar er den første dag i året, og den 31. december er den sidste dag. Hvad med dagene imellem? Den følgende formel returnerer dagen i året, de er gemt i celle A1:

```
=A1-DATO (ÅR (A1) ; 1; 0)
```

Argumentet for dagen er 0 (nul), der spørges på den 0'te dag i den første måned. Funktionen DATO fortolker dette som dagen før den første dag, eller den 31. december i det forgående år i dette eksempel. På samme måde kan negative tal benytte som dag argumentet.

Hvor er funktionen DATO.FORSKEL?

Flere steder i dette notesæt refererer jeg til funktionen DATO.FORSKEL. Du har måske bemærket at funktionen ikke optræder i dialogboksen Indsæt funktion, og den findes ikke på listen Dato og klokkeslæt, og den dukker ikke op i Formel Autoudfyld listen. Derfor skal du altid skrive funktionen manualt, hvis du vil benytte den.

Funktionen DATO.FORSKEL har dens oprindelse i Lotus 1-2-3 og Excel giver adgang til den udelukkede for kompatibilitets årsager. Af forskellige årsager ønsker Microsoft at beholde denne funktion som en hemmelighed. Du vil ikke en gang finde funktionen DATO.FORSKEL i hjælpe filerne, selvom den er tilgængelig i alle versionerne af Excel. Mærkelig nok dukkede DATO.FORSKEL op i hjælpe filerne til Excel 2000 og er ikke set siden.

Funktionen DATO.FORSKEL er smart til at beregne antallet af dage, måneder eller år mellem to datoer. Funktionen har tre argumenter: start_dato, slut_dato og en kode, der repræsenterer den ønskede tidsenhed. Her er et eksempel på en formel der benytter funktionen DATO.FORSKEL (antag at celle A1 og A2 indeholder datoer). Formlen returnerer antallet af hele år mellem de to datoer.

```
=DATO.FORSKEL (A1; A2; "y")
```

Følgende tabel viser de gyldige koder du kan bruge som tredje argument. Du skal sætte citationstegn omkring.

Enheds kode	Returnere
"y"	Antallet af gennemrørte år i perioden.
"m"	Antallet af gennemrørte måneder i perioden.
"d"	Antallet af dage i perioden.
"md"	Forskellen i dage mellem start_dato og slut_dato. Månederne og årene i datoerne ignoreres.
"ym"	Forskellen i måneder mellem start_dato og slut_dato. Månederne og årene i datoerne ignoreres.
"yd"	Forskellen i dage mellem start_dato og slut_dato. Årene i datoerne ignoreres.

Argumentet start_dato skal være før argumentet slut_dato, ellers returnerer funktionen en fejl.

Her er en tilsvarende formel der returnerer dagen i året for den nuværende dato:

```
=IDAG () -DATO (ÅR (IDAG ()) ; 1; 0)
```

Følgende formel returnerer antallet af dage, der er tilbage i året fra en bestemt dato (antag den er i celle A1):

```
=DATO (ÅR (A1) ; 12; 31) -A1
```

Når du indtaster denne formel, vil Excel tilføje datoformatering til cellen. Du er nødt til selv at tilføje en formatering, der ikke er en dato for at se resultatet som et tal.

For at konvertere en bestemt dag i året (f.eks. dag nummer 90 i året) til en faktisk dato i et specifikt år, kan du benytte den formel der følger. Formlet antager at året er gemt i celle A1 og dagen i året er gemt i celle B1:

```
=DATO (A1 ; 1 ; B1)
```

Bestem dagen i ugen

Funktionen UGEDAG accepterer en dato som argument og returnerer et heltal mellem 1 og 7, der korresponderer med dagen i ugen. F.eks. vil følgende formel returnere 2 da den første dag i året 2018 falder på en mandag:

```
=UGEDAG (DATO (2018 ; 1 ; 1) )
```

Funktionen UGEDAG benytter et andet valgfrit argument, der specificerer dag-talsystemet i resultatet. Hvis du angiver 2 som det andet argument, vil funktionen returnere 1 for mandag, 2 for tirsdag osv. Hvis du specificere 3 som det andet argument, vil funktionen returnere 0 for mandag, 1 for tirsdag osv.

Du kan også bestemme ugedagen for en uge i en celle, der indeholder en dato ved at tilføje et brugerdefineret format. En celle der bruger følgende brugerdefineret format viser ugedagen i ugen:

```
dddd
```

Bestem ugen i året

For at bestemme ugen i året for en dato, benyttes funktionen UGE.NR. Den følgende formel returnerer ugenummeret for datoen i celle A1:

```
=UGE . NR (A1)
```

Når du bruger funktionen UGE.NR kan du specificeret et valgfrit andet argument, der indikere typen af uge talsystemet du foretrækker. Det andet argument kan være en af ti værdier, som er beskrevet i hjælpesystemet.

Bestem datoen for den seneste søndag

Du kan benytte følgende formel til at returnerer datoen for den forgående søndag. Hvis den nuværende dag er en søndag vil formelen returnere den øjeblikkelige dato. (Du vil være nødt til at formatere cellen for at vise den som en dato.)

```
=IDAG () -REST (IDAG () -1 ; 7)
```

Du kan justere formelen hvis du vil finde andre dage end søndag. Udskift 1 med et andet tal mellem 2 (for mandag) og 7 (for lørdag).

Bestem den første dag i ugen efter en dato

Den næste formel returnerer den specifikke dag i ugen, der optræder efter en bestemt dato. F.eks. kan du benytte denne formel til at bestemme datoen på den første mandag efter en bestemt dato. Denne formel antager at celle A1 indeholder en dato og celle A2 indeholder et tal mellem 1 og 7 (1 for søndag, 2 for mandag osv.).

```
=A1+A2-UGEDAG (A1) + (A2<UGEDAG (A1) ) *7
```

Hvis celle A1 indeholder 04-08-2017 (en fredag) og celle A2 indeholder 2 (for mandag), vil formelen returnere 07-08-2017. Det er den første mandag efter 04-08-2017.

Bestem den n'te forekomst af en ugedag i en måned

Du kan måske få brug for at bestemme datoen på en bestemt forekomst af en ugedag. Antag f.eks. at dit firmas lønningsdag falder den anden fredag i hver måned, og du har brug for at bestemme alle lønningsdagene i løbet af et år. Den følgende formel udføre denne type af beregning:

```
=DATO (A1 ; A2 ; 1) + A3 - UGEDAG (DATO (A1 ; A2 ; 1) ) + (A4 -  
(A3 >= UGEDAG (DATO (A1 ; A2 ; 1) ) ) ) * 7
```

Formlen i denne sektion antager at

- Celle A1 indeholder et år
- Celle A2 indeholder en måned
- Celle A3 indeholder et nummer på en dag (1 for søndag, 2 for mandag osv.).
- Celle A4 indeholder forekomst tallet (f.eks. 2 for at vælge den anden forekomst af ugedagen specificeret i celle A3).

Hvis du benytter denne formel til at bestemme datoen på den tredje fredag i august 2017, returnere den fredag 18. august 2017.

Hvis værdien i celle A4 overskrider antallet af forekomster for en specifik dag i måneden, vil formelen returnere en dato fra den efterfølgende måned. Hvis du f.eks. forsøger at bestemme den femte fredag i august 2017 (der findes ikke sådan en dato), vil formelen returnere den første fredag i september.

Tæl antallet af forekomster af en ugedag

Du kan benytte følgende formel til at tælle antallet af forekomster af en bestemt ugedag for en specifik måned. Det antages at celle A1 indeholder en dato og celle B1 indeholder en dag (1 for søndag, 2 for mandag osv.). Formlen er en array formel, så du skal indtaste den ved at taste Ctrl+Shift+Enter.

```
{=SUM ( ( UGEDAG (DATO (ÅR (A1) ; MÅNED (A1) ; RÆKKE (INDIREKTE ("1:" & DAG (DATO (ÅR (A1) ;  
MÅNED (A1) +1 ; 0) ) ) ) ) =B1) *1) }
```

Hvis celle A1 indeholder datoen 4 august 2017 og celle B1 indeholder værdien 2 (for mandag), vil formelen returnere 4, som røber at der i august 2017 er fire mandage.

Den forgående array formel beregner året og måneden ved at bruge formlerne ÅR og MÅNED. Du kan forenkle denne formel en smule, hvis du gemmer år og måned i separate celler. Den følgende formel (også en array formel) antager at året findes i celle A1 og måneden i celle A2, og tallet for dagen i celle B1:

```
=SUM ( ( UGEDAG (DATO (A1 ; A2 ; RÆKKE (INDIREKTE ("1:" & DAG (DATO (A1 ; A2 +1 ; 0) ) ) ) ) =B1)  
*1)
```

Du kan læse mere om arrayformler i notesættet Arrays

Figur 5 viser denne formel brugt i et regneark. I dette tilfælde bruger formelen et miks af cellereferencer så du kan kopiere den. F.eks. er formelen i celle C3

```
=SUM ( (UGEDAG (DATO ($B$2; $A3; RÆKKE (INDIREKTE ("1:"&DAG (DATO ($B$2; $A3+1; 0) ) ) ) ) ) =C$1) *1)
```

De øvrige formler benytter funktionen SUM til at beregne antallet dage per måned (i kolonne J) og antallet af dage i ugedage hver måned i året (række 15).

	A	B	C	D	E	F	G	H	I	J	
1			1	2	3	4	5	6	7		
2	År-->	2017	Mandag	Tirsdag	Onsdag	Torsdag	Fredag	Lørdag	Søndag	Måned	
3	1	Januar	5	5	5	4	4	4	4	31	
4	2	Februar	4	4	4	4	4	4	4	28	
5	3	Marts	4	4	4	5	5	5	4	31	
6	4	April	5	4	4	4	4	4	5	30	
7	5	Maj	4	5	5	5	4	4	4	31	
8	6	Juni	4	4	4	4	5	5	4	30	
9	7	Juli	5	5	4	4	4	4	5	31	
10	8	August	4	4	5	5	5	4	4	31	
11	9	September	4	4	4	4	4	5	5	30	
12	10	Oktober	5	5	5	4	4	4	4	31	
13	11	November	4	4	4	5	5	4	4	30	
14	12	December	5	4	4	4	4	5	5	31	
15		Antal dage:	53	52	52	52	52	52	52	365	
16											

Figur 5 Beregning af antallet af hver ugedag i hver måned i et år

Bestem den sidste dag i en måned

For at bestemme datoen for den tilhørende sidste dag i en måned, kan du bruge funktionen DATO. Men du er nødt til at øge måneden med 1, og bruge værdien for dagen 0 (nul). Med andre ord den nul'te dag i den næste måned er den sidste dag i den nuværende måned.

Den følgende formel antager at en dato er gemt i celle A1. Formlen returnerer datoen der korresponderer med den sidste dag i måneden.

```
=DATO (ÅR (A1) ; MÅNED (A1) +1 ; 0)
```

Du kan bruge en variation af denne formel til at bestemme, hvor mange dage, der er i en specifik måned. Formlen der følger returnerer et heltal, der korresponderer med antallet af dage i måneden for datoen i celle A1.

```
=DAG (DATO (ÅR (A1) ; MÅNED (A1) +1 ; 0) )
```

Bestem om år er et skudår

For at afgøre om et bestemt år er et skudår, kan du skrive en formel der afgøre om den 29. dag i februar optræder i februar eller marts. Du kan drage fordel af det faktum at Excels DATO funktion justerer resultatet når du indtaster et ugyldigt argument – f.eks. dag 29 i februar når februar kun indeholder 28 dage.

Den følgende formel returnerer SAND hvis et år i celle A1 er et skud år, og ellers returnerer det FALSK.

```
=HVIS (MÅNED (DATO (A1 ; 2 ; 29) ) =2 ; SAND ; FALSK)
```

Denne funktion returnerer et forkert resultat (SAND) hvis året er 1900. Se afsnittet "Excel skudårsfejl", tidligere i notesættet.

Den følgende formel er en smule mere kompliceret, men identificere korrekt 1900 som et ikke skudsår. Denne formel antager at celle A1 indeholder et årstal.

```
=HVIS (ELLER (REST (A1; 400) =0; OG (REST (A1; 4) =0; REST (A1; 100) <>0) ) ; SAND; FALSK)
```

Bestem en datos kvartal

Til finansielle rapporter vil du opdage at det kan være brugbart at repræsentere information udtrykt som kvartaler. Den følgende formel returnere et heltal mellem 1 og 4, der korresponderer med kvartaler i kalenderen for datoen i celle A1:

```
=RUND.OP (MÅNED (A1) / 3; 0)
```

Denne formel dividerer måneden med 3 og runder derefter op til resultatet.

Konverter et år til romerske tal

Fans af gamle film vil helt sikkert kunne lide denne her. Den følgende formel konverterer året 1960 til dette romertal: MCMLX:

```
=ROMERTAL (1960)
```

Denne funktion konverterer en tekststreng, så du kan ikke udføre beregninger med resultatet.

Endelig! Den længeventet makker funktion blev introduceret i Excel 2013. Brug funktionen ARABISK til at konverterer et romertal til en værdi. Her er et eksempel, som returnere 1960:

```
=ARABISK ("MCMLX")
```

Tidsrelaterede funktioner

Excel inkluderer et antal funktioner der gør det muligt at arbejde med klokkeslæt i dine formler. Denne sektion indeholder eksempler der demonstrere brugen af disse funktioner.

Tabel 5 opsummerer de tilgængelige klokkeslæts funktioner der er i Excel. På samme måde som dato funktionerne omtalt tidligere, kan klokkeslæts funktionerne findes under listefeltet Dato og klokkeslæt via Formler → Funktionsbiblioteket.

Tabel 5: Klokkeslæts relaterede funktioner

Funktion	Beskrivelse
TIME	Returnerer klokkeslættet for en klokkeslætsværdi. Klokkeslættet vises som et heltal mellem 0 og 23.
MINUT	Returnerer minuttallene for en klokkeslætsværdi. Tallet vises som et heltal mellem 0 og 59.
NU	Returnerer serienummeret for den aktuelle dato og det aktuelle klokkeslæt.
SEKUND	Returnerer antallet af sekunder for en tidsværdi. Antallet af sekunder returneres som et heltal mellem 0 (nul) og 59.
TID	Returnerer decimaltallet for et bestemt klokkeslæt.

Funktion	Beskrivelse
TIDSVÆRDI	Returnerer decimaltallet for det klokkeslæt, der repræsenteres af en tekststreng.

Få vist en nuværende tid

Denne formel viser den nuværende tid som et tid serietal (eller et serietal uden den tilhørende dato):

```
=NU() - IDAG()
```

Du er nødt til at formatere cellen med et klokkeslæts format for at se resultatet som en genkendelig tid. Den hurtigste måde at gøre dette på er ved at vælge **Hjem** → **Tal** → **Talformater** og så vælge **Klokkeslæt** i listefeltet.

Du kan også vise tiden kombineret med tekst. Denne formel der følger viser denne tekst: *Det nuværende klokkeslæt er 17:20.*

```
= "Det nuværende klokkeslæt er "&TEKST(NU(); "t:mm")
```

Disse formler opdateres kun når regnearket genberegnes. Tiden kommer fra din computers klokkeslæt, så hvis klokken er forkert, vil formelen returnere et forkert klokkeslæt.

Får vist klokkeslæt ved hjælp af en funktion

Tidligere i notesættet beskrev jeg hvordan man kan indtaste en tidsværdi i en celle: bare indtast i en celle, men vær sikker på at inkludere mindst et kolon (:). Du kan også oprette et klokkeslæt ved at bruge funktionen TID. Den følgende funktion vil f.eks. returnere et klokkeslæt sammensat af timen i celle A1, minuttet i celle B1 og sekundet i celle C1:

```
=TID(A1;B1;C1)
```

På samme måde som DATO funktionen, accepterer funktionen TID ugyldige argumenter og justerer dem efterfølgende. Den følgende formel bruger f.eks. 80 som argumentet for minutter og returnerer 10:20:15. De 80 minutter lægges ganske enkelt til timen, med 20 minutter i overskud.

```
=TID(9;80;15)
```

Hvis du indtaster en værdi større end 24 som det første argument i funktionen TID, vil resultatet sikkert ikke være som du havde forventet. Logisk vil en sådan formel som den der følger producere et dato/klokkeslæt serietal på 1,041667 (det er en dag og en time):

```
=TID(25;0;0)
```

Faktisk er formelen ækvivalent til den følgende:

```
=TID(1;0;0)
```

Du kan også bruge funktionen DATO samme med funktionen TID i en enkelt celle. Formlen der følger generer en dato og et klokkeslæt med et serietal på 42833,69 – som repræsenterer kl. 16:30 den 8. august 2017.

```
=DATO (2017; 4; 8) +TID (16; 30; 0)
```

Når du indtaster den forgående formel, vil Excel formatere cellen så du kun ser datoen. For også at se klokkeslættet, skal du skifte talformateringen til en der viser dato og klokkeslæt.

Funktionen TIDSVÆRDI konverterer en tekststreng der ligner et klokkeslæt til et klokkeslæt serietal. Denne formel returnerer 0,65625, som er klokkeslæt serietallet for 15:45:

```
=TIDSVÆRDI ("15:45")
```

Beregn forskellen mellem to klokkesæt

Da klokkeslæt repræsenteres ved et serietal, kan du trække det tidlige klokkeslæt fra det senere klokkeslæt for at få forskellen. F.eks. hvis celle A2 indeholder 5:30:00 og celle B2 indeholder 14:00:00, vil følgende formel returnere 08:30:00 (en forskel på otte timer og 30 minutter):

```
=B2-A2
```

Hvis subtraktionen giver en negativ værdi, bliver det dog en ugyldig værdi: Excel viser en række af havlåger (####) fordi et klokkeslæt uden dato har et serietal på 0. Et negativ klokkeslæt resultat i et negativt serietal, som ikke kan vises – du kan dog stadig benytte den beregnet værdi i en anden formel.

Hvis rækkefølgen på forskellen i klokkeslættene ikke har betydning, kan du bruge funktionen ABS som returnere en absolut værdi for forskellen:

```
=ABS (B2-A2)
```

Det "negative klokkeslæts" problem optræder ofte når der beregnes en overløbet tid - f.eks. beregning af antallet timer der er arbejdet med en start og slut klokkeslæt. Dette er ikke noget problem hvis de to klokkeslæt falder på den samme dag. Hvis slut klokkeslættet finder sted efter midnat, vil resultat give en ugyldig negativ klokkeslæt. Du kan f.eks. starte arbejdet kl. 22:00 og slutte arbejdet næste dag kl. 6:00. Figur 6 viser et arbejdschema, der beregner antallet af timer der er arbejdet. Som du kan se den dag hvor arbejdet løber over midnat er der et problem.

	A	B	C
1	Start	Slut	Arbejdstimer
2	08:00	17:30	09:30
3	22:00	06:00	#####
4	09:00	16:30	07:30
5	11:30	19:45	08:15
6	06:15	11:00	04:45
7			

Figur 6 Beregning af antallet af arbejdstimer returnerer en fejl hvis det løber over midnat.

Brug af funktionen ABS (til beregning af den absolutte værdi) er ikke en mulighed i dette tilfælde, da det vil returnere det forkerte resultat (16 timer). Den følgende formel virker dog:

```
=HVIS (B2<A2;B2+1;B2) -A2
```

Og en formel der er endnu simplere kan klare arbejdet:

```
=REST (B2-A2;1)
```

Negative klokkeslæt er tilladt hvis regnearket benytter datosystemet 1904. For at skifte til datosystemet 1904 vælges Filer → Indstillinger og under fanen Avanceret navigeres frem til Brug 1904-daosystem. Sæt et flueben ved siden af indstillingen Brug 1904-daosystem. Vær opmærksom på at når du skifter et regnearks datosystem, vil regnearkets datoer, hvis der benytte datoer, være fire år ved siden af.

Sammentælling af klokkeslæt der overstiger 24 timer

Mange bliver overrasket over at opdage at når du summerer en serie af klokkeslæt, der overstiger 24 timer, da vil Excel ikke vise den korrekte total. Figur 8 viser et eksempel. Området B2:B8 indeholder klokkeslæt, der repræsenterer timer og minutter der arbejdes hver dag. Formlen i celle B9 er

```
=SUM (B2:B9)
```

	A	B	C
1	Dag	Arbejdstimer	
2	Mandag	08:30	
3	Tirsdag	08:00	
4	Onsdag	09:00	
5	Torsdag	09:30	
6	Fredag	04:15	
7	Lørdag	02:30	
8	Søndag	00:00	
9	Antal timer	17:45	
10			

Figur 7 Celleformatering der ikke er korrekt får totalen til at vises forkert.

Som du kan se returnerer formelen en forkert total (17 timer og 45 minutter). Totalen burde være 41 timer og 45 minutter. Problemet er at formelen viser totalen som en dato/tids serietal på 1,7395833, men celle formateringen viser ikke dato delen i dato/tid. Svaret er forkert da celle B9 har et forkert talformat.

For at se en tid, der overstiger 24 timer, skal du ændre cellens talformatering, med kantede parenteser omkring time delen i formateringsstrengen. Tilføjelse at talformatet her til celle B9 viser summe korrekt:

```
[h] :mm
```

Figur 9 viser at andet eksempel på et arbejdsskema der manipulerer klokkeslæt. Dette regneark holder styr på antal timer der arbejdes i løbet af en uge (almindelige timer og overarbejdstimer).

Ugens startdato vises i celle D5, og formlerne i kolonne C udfylder datoerne for dagene i ugen. Klokkeslættene vises i celleområdet D8:G14, og formlerne i kolonne H beregner antallet af timer der arbejdes hver dag. F.eks. er formelen i H8

```
=HVIS (E8<D8;E8+1-D8;E8-D8)+HVIS (G8<F8;G8+1-G8;G8-F8)
```

Den første del af formelen trækker klokkeslættet i kolonne D fra klokkeslættet i kolonne E for at få antallet af arbejdstimer før frokost. Den anden del trækker klokkeslættet i kolonne F fra klokkeslættet i kolonne G for at få antallet af arbejdstimer efter frokost. Jeg bruger HVIS funktionen for at imødegå skiftet der sker når klokkeslættene spænder over midnat – f.eks. hvis en ansat starter arbejdet kl. 22.00 og begynder sin frokost kl. 2.00. Uden HVIS funktionen vil formelen returnere et negativt resultat.

Den følgende formel i celle E17 beregner den ugentlige total ved at summere den daglige total i kolonne H:

```
=SUM (H8:H14)
```

	A	B	C	D	E	F	G	H	I	J
1										
2										
3										
4										
5										
6										
7										
8										
9										
10										
11										
12										
13										
14										
15										
16										
17										
18										
19										
20										
21										
22										
23										
24										

Arbejdsskema							
Navn		Peter Jensen					
Afdeling		Kontor					
Startdato		18-09-2017					
Ugedag	Dato	Start Arbejde	Tid ud (frokost)	Tid ind (frokost)	Slut Arbejde	Total Timer	Uge Timer
mandag	18-09-2017	08:00	12:00	13:00	18:00	9:00	09:00
tirsdag	19-09-2017	10:00	14:00	14:30	20:00	9:30	18:30
onsdag	20-09-2017	09:00	12:00	13:00	19:30	9:30	28:00
torsdag	21-09-2017	08:00	12:00	13:00	18:30	9:30	37:30
fredag	22-09-2017					0:00	37:30
lørdag	23-09-2017	21:00			1:00	4:00	41:30
søndag	24-09-2017					0:00	41:30
Uge total							
Timer total					41:30		
Almindelige timer					37:00		
Overarbejdstimer					4:30		

Overtid	37:00
---------	-------

Figur 8 En ansats timeregnskab.

Dette regneark antager at timer der overstiger 37 timer i en uge betragtes som overarbejdstimer. Dette regneark indeholder en celle, der hedder *Overarbejdstimer* (celle C23), der indeholder 37:00. Hvis dit standard regneark indeholder noget andet end 37 timer, kan du ændre cellen *Overarbejdstimer*.

Den følgende formel (i celle E18) beregner det almindelige (ikke overarbejdstimer) antal timer. Denne formel returnerer den mindste af to værdier: det samlede antal timer eller overarbejdstimerne.

```
=MIN (E17;Overarbejdstimer)
```

Den sidste formel, i celle E19, trækker det almindelige antal timer fra det samlede antal timer for at få overarbejdstimerne:

```
=E17-E18
```

Klokkeslættene i E17:E19 kan vises klokkeslæt, der overskrider 24 timer, så cellerne benytter en brugerdefineret formatering:

```
[h] :mm
```

Konverter fra militær klokkeslæt

Militær klokkeslæt angives ved fircifrede tal fra 0000 til 2359. F.eks. vil 01:00 skrives som 0100, og 15:30 skrives som 1530. Den følgende formel konvertere et sådan tal (antag det står i celle A1) til et standard klokkeslæt:

```
=TIDSVÆRDI (VENSTRE (A1; 2) &" : "&HØJRE (A1; 2) )
```

Formlen returnerer et forkert resultat hvis indholdet i celle A1 ikke indeholder fire cifre. Den følgende formel retter dette problem og returner et gyldigt klokkeslæt for ethvert militært klokkeslæt fra 0 til 2359:

```
=TIDSVÆRDI (VENSTRE (TEKST (A1; "0000") ; 2) &" : "&HØJRE (A1; 2) )
```

Den følgende simple formel der bruger funktionen TEKST returnerer en formateret streng og bruger derefter funktionen TIDSVÆRDI til at udtrykke resultatet som et klokkeslæt:

```
=TIDSVÆRDI (TEKST (A1; "00\ :00" ) )
```

Konverter decimal timer, minutter eller sekunder til et klokkeslæt

For at konvertere decimal timer, divideres decimal timerne med 24. F.eks. hvis en celle A1 indeholder 9,25 (repræsenterer timerne) vil denne formel returnere 09:15:00 (9 timer, 15 minutter):

```
=A1/24
```

For at konvertere decimal minutter til et klokkeslæt, divideres decimal minutterne med 1.440 (antallet af minutter på en dag). F.eks. hvis cellen A1 indeholder 500 (antallet af minutter), vil følgende formel returnere 08:20:00 (8 timer, 20 minutter):

```
=A1/1440
```

For at konvertere decimal sekunder til et klokkeslæt, divideres decimal sekunderne med 86.400 (antallet af sekunder på en dag). F.eks. hvis celle A1 indeholder 65.000 (antallet af sekunder), vil følgende formel returnere 18:03:20 (18 timer, 3 minutter og 20 sekunder):

```
=A1/86400
```

Tilføj timer, minutter eller sekunder til et klokkeslæt

Du kan bruge funktionen TID til at tilføje et antal timer, minutter eller sekunder til et klokkeslæt. Antag f.eks. at celle A1 indeholder et klokkeslæt. Den følgende formel lægger to timer og 30 minutter til dette klokkeslæt og viser resultatet:

```
=A1+TID (2; 30; 0)
```

Du kan bruge funktionen TID til at udfylde et celleområde med forøget klokkeslæt. Figur 10 viser et regneark med en serie af klokkeslæt i ti-minutters intervaller. Celle A1 indeholder et klokkeslæt der er indtastet direkte. Celle A2 indeholder følgende formel, som kopieres ned over kolonnen:

A2						
=A1+TID(0;10;0)						
	A	B	C	D	E	F
1	08:00					
2	08:10					
3	08:20					
4	08:30					
5	08:40					
6	08:50					
7	09:00					
8	09:10					
9	09:20					
10	09:30					
11	09:40					
12	09:50					
13						

Figur 9 Benyt en formel til at oprette en serie af tidsforøgelser.

Konverter mellem tidszoner

Det kan være du modtager et regneark, der indeholder datoer og klokkeslæt i Greenwich Mean Time (GMT, nogle gange refereret til som *Zulu tid*), og det kan være du har brug for at konvertere disse værdier til lokal tid. For at konvertere datoer og klokkeslæt til lokale tider, har du brug for at bestemme forskellen mellem de to tidszoner. F.eks. for at konvertere GMT tider til København tider, er konverterings faktoren +1.

Du kan ikke bruge funktionen Tid med et negativt argument, du er derfor nødt til at angribe problemet fra en anden vinkel. En time er lig med 1/24 dag, så du kan dividere tidsomregningsfaktoren med 24 og derefter lægge den til klokkeslættet.

Figur 11 viser et regneark der konvertere datoer til klokkeslæt (udtrykt i GMT) til lokale klokkeslæt. Celle B1 indeholder time tidsomregningsfaktoren (+1 for København). Formlen i B4, som kopieres ned over kolonnen, er

```
=A4+ ($B$1/24)
```

Denne formel lægger x timer til datoen og klokkeslættet i kolonne A. Hvis celle B1 indeholder et negativt klokkeslæt, trækkes værdien fra datoen og klokke i kolonne A. Bemærk at i nogle tilfælde vil dette også have en indflydelse på datoen.

B4				=A4+(\$B\$1/24)
	A	B	C	
1	Tidsomregningsfaktoren		1 time	
2				
3	GMT	Lokaltid		
4	01-09-2017 01:00	01-09-2017 02:00		
5	01-09-2017 01:30	01-09-2017 02:30		
6	01-09-2017 02:00	01-09-2017 03:00		
7	01-09-2017 02:30	01-09-2017 03:30		
8	01-09-2017 03:00	01-09-2017 04:00		
9	01-09-2017 03:30	01-09-2017 04:30		
10	01-09-2017 04:00	01-09-2017 05:00		
11	01-09-2017 04:30	01-09-2017 05:30		
12	01-09-2017 05:00	01-09-2017 06:00		
13	01-09-2017 05:30	01-09-2017 06:30		
14	01-09-2017 06:00	01-09-2017 07:00		
15	01-09-2017 06:30	01-09-2017 07:30		
16	01-09-2017 07:00	01-09-2017 08:00		
17	01-09-2017 07:30	01-09-2017 08:30		
18	01-09-2017 08:00	01-09-2017 09:00		
19				

Figur 10 Dette regneark konverterer datoer og klokkeslæt mellem tidszoner.

Afrunding af klokkeslæt

Du kan få brug for en formel der afrunder et klokkeslæt til en bestemt værdi. F.eks. kan du få brug for at indtaste et klokkeslæt i dit firmas arbejdsskema til nærmeste 15 minutter. Dette afsnit præsenterer dig for forskellige måder at afrunde et klokkeslæt.

Den følgende formel afrunder klokkeslættet i celle A1 til nærmeste minut:

```
=AFRUND (A1*1440;0)/1440
```

Formlen ganger klokkeslættet med 1440 (for at få det samlede antal minutter). Denne værdi videreføres til funktionen AFRUND og resultatet divideres med 1440. Hvis f.eks. celle A1 indeholder 12:40:35 vil formelen returnere 12:41:00.

Den følgende formel efterligner dette eksempel, men afrunder klokkeslættet i celle A1 til nærmeste time:

```
=AFRUND (A1*24;0)/24
```

Hvis celle A1 indeholder 17:21:35 vil formelen returnere 17:00:00.

Den følgende formel afrunder klokkeslættet i celle A1 til nærmeste 15 minutter (en kvart time):

```
=AFRUND (A1*24/0,25;0)*(0,25/24)
```

I denne formel repræsenterer 0,25 timedelen. For at afrunde til nærmeste 30 minutter, erstattes 0,25 med 0,5, som i følgende formel:

```
=AFRUND (A1*24/0,5;0)*(0,5/24)
```

Arbejdet med tidsværdier

Ind imellem kan du komme til at arbejde med tidsværdier, der ikke repræsenterer en faktisk tid i løbet af dagen. F.eks. skal du oprette en liste med sluttider i et motionsløb eller måle hvor lang tid du løber hver

dag. Sådanne tider repræsenterer ikke et klokkeslæt i løbet af dagen. En værdi repræsenterer snarere tidspunktet for en begivenhed (i timer, minutter og sekunder). Tiden det tager at udføre en test, kan f.eks. være 35 minutter og 45 sekunder. Du kan indtaste denne værdi en i celle som

00:35:45

Excel opfatter denne indtastning som klokkeslættet 00:35:45, som fungerer fint (bare vær sikker på du formaterer celle, så det ser ud som du vil have). Når du indtaster en sådan tid, der ikke har en time del, skal du inkludere mindst et nul for timen. Hvis du undlader det foranstillet nul for den manglende time, vil Excel opfatte din indtastning som 35 timer og 45 minutter.

Figur 12 viser et eksempel på et regneark, der er sat op så det holder styr på en persons løbeaktiviteter. Kolonne A indeholder simple datoer. Kolonne B indeholder afstanden i kilometer. Kolonne C indeholder tiden det tog at løbe distancen. Kolonne D indeholder formler der beregner hastigheden i kilometer per time. Formlen i celle D2 er f.eks.

=B2 / (C2*24)

	A	B	C	D	E	F	G
1	Dato	Afstand	Tid	Fart km/t	Tempo (min/km)	Afstand Kommuleret	Tid Kommuleret
2	01-09-2017	3,00	00:18:54	9,52	6,30	3,00	00:18:54
3	02-09-2017	3,00	00:17:40	10,19	5,89	6,00	00:36:34
4	03-09-2017	4,00	00:21:30	11,16	5,38	10,00	00:58:04
5	04-09-2017	3,00	00:15:20	11,74	5,11	13,00	01:13:24
6	05-09-2017	4,80	00:25:05	11,48	5,23	17,80	01:38:29
7	06-09-2017	6,00	00:31:06	11,58	5,18	23,80	02:09:35
8	07-09-2017	7,60	00:41:06	11,09	5,41	31,40	02:50:41
9	08-09-2017	10,00	01:09:00	8,70	6,90	41,40	03:59:41
10	09-09-2017	8,00	00:45:10	10,63	5,65	49,40	04:44:51
11	10-09-2017	6,00	00:29:06	12,37	4,85	55,40	05:13:57
12	11-09-2017	11,00	01:08:30	9,64	6,23	66,40	06:22:27
13							

Figur 11 Dette regneark benytter klokkeslæt, der ikke er tilknyttet et bestemt tidspunkt på dagen.

Kolonne E indeholder formler der beregner tempoet i minutter per kilometer. F.eks. er formelen i E2

= (C2*60*24) / B2

Kolonne F og G indeholder formler der beregner den kumulerede afstand (ved hjælp af kolonne B9 og den kumulerede tid (ved hjælp af kolonne C). Cellerne i kolonne G er formateret ved hjælp af følgende talformatering (som tillader visning af klokkeslæt der overstiger 24 timer):

[hh]:mm:ss